INFRASTRUCTURE DEPLOYMENT WITH TERRAFORM

PHASE 1

🔷 Azure Virtual Desktop – LAB: Deploying AVD Environment from Scratch

DevOps & Azure Provisioning with Terraform + Ansible + Microsoft Entra Connect Sync
Author: Lucas Román Vercellini
Goal: Deploy and configure a hybrid Windows + Azure environment with Domain Controller, identity sync, security hardening, base software, and secure access.

Term	Definition
Terraform	IaC tool that allows you to define, provision, and version infrastructure using declarative syntax (HCL).
Resource Group	Logical container for Azure resources, simplifying management, access, and cost control.
Provider	Terraform plugin that understands how to interact with a specific cloud API (e.g. `azurerm` for Azure).
State File	File that stores the current infrastructure state, essential for consistent `plan` and `apply`.
NSG (Network Security Group)	Firewall rules at the network level to control inbound/outbound traffic to subnets or NICs.
Dynamic IP	IP address assigned automatically by Azure, which may change when the resource is stopped/restarted.

🔧 Project Structure

```
azure-avd-lab/
├── terraform/
│   ├── main.tf
│   ├── variables.tf
│   ├── outputs.tf
│   ├── providers.tf
│   ├── lab.tfvars
│   └── bastion.tf (opcional, luego)
├── ansible/
│   ├── playbook.yml
│   ├── hosts.ini
│   └── roles/
│       └── windows_config/
│           ├── tasks/
│           │   ├── main.yml
│           │   └── hardening.yml
├── docs/
│   ├── README.md ← documentación principal en inglés
│   └── screenshots/ ← capturas paso a paso
└── .gitignore
```

📘 PHASE 1 – Infrastructure Deployment with Terraform

✅ Required Information to Start

Azure Account Data:

1. **Subscription ID**
   `az account show --query id -o tsv`
2. **Tenant ID**
   `az account show --query tenantId -o tsv`
3. **Resource Group**
   Defined via Terraform: `rg-avd-lab`
4. **Region / Location**
   Recommended: `East US` (verified for free accounts)
5. **Base Resource Name**
   Convention: `avd-lab` (used as prefix for VM, NSG, IP, etc.)
6. **VM Size**
   `Standard_B2ms` (2 vCPU, 8 GB RAM) → suitable for Windows & testing
7. **Windows VM Base Image**
   - Windows 11 Enterprise multi-session:
     - publisher = "MicrosoftWindowsDesktop"
     - offer = "windows-11"
     - sku = "win11-21h2-ent"
     - version = "latest"
8. **Local Administrator Credentials**
   - Username: azureuser
   - Password: defined in lab.tfvars
   -

⚠️ Important: Two Separate Environments Are Used

Este laboratorio utiliza **DOS entornos diferentes**:

- 🖥️ Local Environment (VirtualBox) – Domain Controller (lab.local)
  Local VM with Windows Server 2022
- ☁️ Azure Environment – Target VM deployed with Terraform
  VM password is set in lab.tfvars

Both environments are synchronized via Microsoft Entra Connect in Phase 3.

⚙️ Variable Configuration

✅ variables.tf

```
variable "location" {
 description = "Azure region where resources will be deployed"
 type        = string
 default     = "East US"
}
```

```
variable "resource_group_name" {
  description = "Name of the resource group"
  type        = string
  default     = "rg-avd-lab"
}

variable "vm_admin_username" {
  description = "Admin username for the Windows VM"
  type        = string
  default     = "azureuser"
}

variable "vm_admin_password" {
  description = "Admin password for the Windows VM"
  type        = string
  sensitive   = true
}

variable "vm_name" {
  description = "Name of the virtual machine"
  type        = string
  default     = "vm-avd-lab"
}

variable "vm_size" {
  description = "Azure VM size"
  type        = string
  default     = "Standard_B2ms"
}
```

✅ **lab.tfvars**

```
location = "East US"
vm_admin_password = "P@ssw0rd123456!"  # Cambiar por uno real y fuerte
```

🧩 **Terraform Configuration Files**

✅ **providers.tf**

```
terraform {
  required_providers {
    azurerm = {
      source  = "hashicorp/azurerm"
      version = "~> 3.79.0"
    }
  }
  required_version = ">= 1.4.0"
}
```

```
provider "azurerm" {
  features {}
  skip_provider_registration = true
}
```

✅ main.tf

```
resource "azurerm_resource_group" "main" {
  name     = var.resource_group_name
  location = var.location
}

resource "azurerm_virtual_network" "main" {
  name                = "vnet-avd"
  address_space       = ["10.0.0.0/16"]
  location            = azurerm_resource_group.main.location
  resource_group_name = azurerm_resource_group.main.name
}

resource "azurerm_subnet" "main" {
  name                 = "subnet-avd"
  resource_group_name  = azurerm_resource_group.main.name
  virtual_network_name = azurerm_virtual_network.main.name
  address_prefixes     = ["10.0.1.0/24"]
}

resource "azurerm_network_security_group" "main" {
  name                = "nsg-avd"
  location            = azurerm_resource_group.main.location
  resource_group_name = azurerm_resource_group.main.name

  security_rule {
    name                       = "Allow-RDP-CustomPort"
    priority                   = 1001
    direction                  = "Inbound"
    access                     = "Allow"
    protocol                   = "Tcp"
    source_port_range          = "*"
    destination_port_range     = "49999"
    source_address_prefix      = "Internet"
    destination_address_prefix = "*"
  }

  security_rule {
    name                       = "Allow-HTTP"
    priority                   = 1002
    direction                  = "Inbound"
    access                     = "Allow"
```

```
  protocol                = "Tcp"
  source_port_range       = "*"
  destination_port_range  = "80"
  source_address_prefix   = "*"
  destination_address_prefix = "*"
 }
}

resource "azurerm_public_ip" "main" {
 name            = "pip-avd"
 location        = azurerm_resource_group.main.location
 resource_group_name = azurerm_resource_group.main.name
 allocation_method  = "Dynamic"
 sku             = "Basic"
}

resource "azurerm_network_interface" "main" {
 name            = "nic-avd"
 location        = azurerm_resource_group.main.location
 resource_group_name = azurerm_resource_group.main.name

 ip_configuration {
  name                          = "ipconfig1"
  subnet_id                     = azurerm_subnet.main.id
  private_ip_address_allocation = "Dynamic"
  public_ip_address_id          = azurerm_public_ip.main.id
 }
}

resource "azurerm_windows_virtual_machine" "main" {
 name              = var.vm_name
 resource_group_name  = azurerm_resource_group.main.name
 location          = azurerm_resource_group.main.location
 size              = var.vm_size
 admin_username      = var.vm_admin_username
 admin_password      = var.vm_admin_password
 network_interface_ids = [azurerm_network_interface.main.id]

 os_disk {
  caching          = "ReadWrite"
  storage_account_type = "Standard_LRS"
 }

 source_image_reference {
  publisher = "MicrosoftWindowsDesktop"
  offer     = "windows-11"
  sku       = "win11-21h2-ent"
  version   = "latest"
```

```
  }

  provision_vm_agent      = true
  enable_automatic_updates = true
  timezone                = "W. Europe Standard Time"

  tags = {
    environment = "lab"
    owner       = "Lucas"
  }
}

resource "azurerm_network_interface_security_group_association" "main" {
  network_interface_id      = azurerm_network_interface.main.id
  network_security_group_id = azurerm_network_security_group.main.id
}
```

✅ outputs.tf

```
output "resource_group_name" {
  description = "The name of the Resource Group"
  value       = azurerm_resource_group.main.name
}

output "vm_name" {
  description = "The name of the deployed Windows virtual machine"
  value       = azurerm_windows_virtual_machine.main.name
}

output "public_ip_address" {
  description = "The public IP address of the VM (for RDP or Bastion access)"
  value       = azurerm_public_ip.main.ip_address
}

output "rdp_port" {
  description = "Custom RDP port configured in the NSG"
  value       = "49999"
}
```

🚀 **Initialization & Deployment**

1. Verify Azure Connection

```
az account show
```

Si la información no es correcta o da error, conectar o reconectar:

```
az login
# o para conectar dispositivo:
az login --use-device-code
```

2. Register Required Providers(critital)

⚠️ **Setup the providers manually:**

az provider register --namespace Microsoft.Network

az provider register --namespace Microsoft.Compute

Verify registration:

az provider show --namespace Microsoft.Compute --query "registrationState"

az provider show --namespace Microsoft.Network --query "registrationState"

3. Git Ignore Settings

Create .gitignore in your project directory:

.terraform/

*.tfstate

*.tfstate.backup

*.tfvars

*.log

*.exe

4. Plan and Apply

If provider errors occur:

Clean init if necesary

rm -rf .terraform .terraform.lock.hcl

terraform init



Verify configuration:

terraform plan -var-file="lab.tfvars"



If everything is ok, then apply:

terraform apply -var-file="lab.tfvars"



✅ Outputs and Final Check

Example after successful apply:

Outputs:

public_ip_address = "20.125.76.22"

rdp_port = "49999"

resource_group_name = "rg-avd-lab"

vm_name = "vm-avd-lab"

📌 Final Lab Overview

- Region: East US
- Subscription ID: *<your subscription ID>*
- Tenant: Default Directory

Created Resources:

- Resource Group: rg-avd-lab
- Virtual Network: vnet-avd
- Subnet: subnet-avd
- NSG: nsg-avd (with RDP only from port 49999)
- NIC: nic-avd
- Windows VM: vm-avd-lab (dynamic public IP)

✅ Infrastructure provisioned successfully